

## NAG Library Routine Document

### F07QWF (ZSPTRI)

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

#### 1 Purpose

F07QWF (ZSPTRI) computes the inverse of a complex symmetric matrix  $A$ , where  $A$  has been factorized by F07QRF (ZSPTRF), using packed storage.

#### 2 Specification

```
SUBROUTINE F07QWF (UPLO, N, AP, IPIV, WORK, INFO)
  INTEGER          N, IPIV(*), INFO
  COMPLEX (KIND=nag_wp) AP(*), WORK(N)
  CHARACTER(1)    UPLO
```

The routine may be called by its LAPACK name *zsptri*.

#### 3 Description

F07QWF (ZSPTRI) is used to compute the inverse of a complex symmetric matrix  $A$ , the routine must be preceded by a call to F07QRF (ZSPTRF), which computes the Bunch–Kaufman factorization of  $A$ , using packed storage.

If UPLO = 'U',  $A = PUDU^T P^T$  and  $A^{-1}$  is computed by solving  $U^T P^T X P U = D^{-1}$ .

If UPLO = 'L',  $A = PLDL^T P^T$  and  $A^{-1}$  is computed by solving  $L^T P^T X P L = D^{-1}$ .

#### 4 References

Du Croz J J and Higham N J (1992) Stability of methods for matrix inversion *IMA J. Numer. Anal.* **12** 1–19

#### 5 Parameters

- 1: UPLO – CHARACTER(1) *Input*  
*On entry:* specifies how  $A$  has been factorized.  
 UPLO = 'U'  
 $A = PUDU^T P^T$ , where  $U$  is upper triangular.  
 UPLO = 'L'  
 $A = PLDL^T P^T$ , where  $L$  is lower triangular.  
*Constraint:* UPLO = 'U' or 'L'.
- 2: N – INTEGER *Input*  
*On entry:*  $n$ , the order of the matrix  $A$ .  
*Constraint:*  $N \geq 0$ .
- 3: AP(\*) – COMPLEX (KIND=nag\_wp) array *Input/Output*  
**Note:** the dimension of the array AP must be at least  $\max(1, N \times (N + 1)/2)$ .  
*On entry:* the factorization of  $A$  stored in packed form, as returned by F07QRF (ZSPTRF).

*On exit:* the factorization is overwritten by the  $n$  by  $n$  matrix  $A^{-1}$ .

More precisely,

if UPLO = 'U', the upper triangle of  $A^{-1}$  must be stored with element  $A_{ij}$  in  $AP(i + j(j - 1)/2)$  for  $i \leq j$ ;

if UPLO = 'L', the lower triangle of  $A^{-1}$  must be stored with element  $A_{ij}$  in  $AP(i + (2n - j)(j - 1)/2)$  for  $i \geq j$ .

4: IPIV(\*) – INTEGER array *Input*

**Note:** the dimension of the array IPIV must be at least  $\max(1, N)$ .

*On entry:* details of the interchanges and the block structure of  $D$ , as returned by F07QRF (ZSPTRF).

5: WORK(N) – COMPLEX (KIND=nag\_wp) array *Workspace*

6: INFO – INTEGER *Output*

*On exit:* INFO = 0 unless the routine detects an error (see Section 6).

## 6 Error Indicators and Warnings

INFO < 0

If INFO =  $-i$ , argument  $i$  had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

Element  $\langle value \rangle$  of the diagonal is exactly zero.  $D$  is singular and the inverse of  $A$  cannot be computed.

## 7 Accuracy

The computed inverse  $X$  satisfies a bound of the form

if UPLO = 'U',  $|DU^T P^T X P U - I| \leq c(n)\epsilon(|D||U^T|P^T|X|P|U| + |D||D^{-1}|)$ ;

if UPLO = 'L',  $|DL^T P^T X P L - I| \leq c(n)\epsilon(|D||L^T|P^T|X|P|L| + |D||D^{-1}|)$ ,

$c(n)$  is a modest linear function of  $n$ , and  $\epsilon$  is the *machine precision*.

## 8 Parallelism and Performance

F07QWF (ZSPTRI) is not threaded by NAG in any implementation.

F07QWF (ZSPTRI) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

## 9 Further Comments

The total number of real floating-point operations is approximately  $\frac{8}{3}n^3$ .

The real analogue of this routine is F07PJF (DSPTRI).

## 10 Example

This example computes the inverse of the matrix  $A$ , where

$$A = \begin{pmatrix} -0.39 - 0.71i & 5.14 - 0.64i & -7.86 - 2.96i & 3.80 + 0.92i \\ 5.14 - 0.64i & 8.86 + 1.81i & -3.52 + 0.58i & 5.32 - 1.59i \\ -7.86 - 2.96i & -3.52 + 0.58i & -2.83 - 0.03i & -1.54 - 2.86i \\ 3.80 + 0.92i & 5.32 - 1.59i & -1.54 - 2.86i & -0.56 + 0.12i \end{pmatrix}.$$

Here  $A$  is symmetric, stored in packed form, and must first be factorized by F07QRF (ZSPTRF).

### 10.1 Program Text

```

Program f07qwfe

!      F07QWF Example Program Text

!      Mark 25 Release. NAG Copyright 2014.

!      .. Use Statements ..
Use nag_library, Only: nag_wp, x04ddf, zsptrf, zsptri
!      .. Implicit None Statement ..
Implicit None
!      .. Parameters ..
Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
Integer                     :: i, ifail, info, j, n
Character (1)               :: uplo
!      .. Local Arrays ..
Complex (Kind=nag_wp), Allocatable :: ap(:), work(:)
Integer, Allocatable        :: ipiv(:)
Character (1)               :: clabs(1), rlabs(1)
!      .. Executable Statements ..
Write (nout,*) 'F07QWF Example Program Results'
!      Skip heading in data file
Read (nin,*)
Read (nin,*) n

Allocate (ap(n*(n+1)/2),work(n),ipiv(n))

!      Read A from data file

Read (nin,*) uplo
If (uplo=='U') Then
  Read (nin,*)((ap(i+j*(j-1)/2),j=i,n),i=1,n)
Else If (uplo=='L') Then
  Read (nin,*)((ap(i+(2*n-j)*(j-1)/2),j=1,i),i=1,n)
End If

!      Factorize A
!      The NAG name equivalent of zsptrf is f07qrf
Call zsptrf(uplo,n,ap,ipiv,info)

Write (nout,*)
Flush (nout)
If (info==0) Then

!      Compute inverse of A
!      The NAG name equivalent of zsptri is f07qwf
Call zsptri(uplo,n,ap,ipiv,work,info)

!      Print inverse

!      ifail: behaviour on error exit
!      =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
ifail = 0
Call x04ddf(uplo,'Nonunit',n,ap,'Bracketed','F7.4','Inverse', &
  'Integer',rlabs,'Integer',clabs,80,0,ifail)

```

```

Else
  Write (nout,*) 'The factor D is singular'
End If

End Program f07qwfe

```

## 10.2 Program Data

F07QWF Example Program Data

```

4                                     :Value of N
'L'                                   :Value of UPLO
(-0.39,-0.71)
( 5.14,-0.64) ( 8.86, 1.81)
(-7.86,-2.96) (-3.52, 0.58) (-2.83,-0.03)
( 3.80, 0.92) ( 5.32,-1.59) (-1.54,-2.86) (-0.56, 0.12) :End of matrix A

```

## 10.3 Program Results

F07QWF Example Program Results

```

Inverse
          1          2          3          4
1  (-0.1562,-0.1014)
2  ( 0.0400, 0.1527) ( 0.0946,-0.1475)
3  ( 0.0550, 0.0845) (-0.0326,-0.1370) (-0.1320,-0.0102)
4  ( 0.2162,-0.0742) (-0.0995,-0.0461) (-0.1793, 0.1183) (-0.2269, 0.2383)

```

---